Module 15 Matrices van polynomen

Onderwerp	Numerieke berekeningen; Bézout-identiteit en Smith-normaalvorm
Voorkennis	Lineaire algebra.
Expressies	`SmithForm, randpoly`
Bibliotheken	`LinearAlgebra`
Zie ook	Module 11, 13, 14, 27.

In deze module behandelen we enige voorbeelden van berekeningen met matrices waarvan de elementen polynomen zijn in plaats van getallen. Dit soort matrices worden vaak gebruikt in de Wiskundige Systeemtheorie.[39]

15.1 Numerieke berekeningen met polynoom-matrices

De elementen van polynoom-matrices zijn *functies*. Hierdoor kunnen we niet meer zonder meer spreken over bijvoorbeeld de rang van de matrix; deze zal afhankelijk zijn van de variabele. Zie bijvoorbeeld ook §14.3, waar de (dimensie van) de oplossingsverzameling van een stelsel vergelijkingen afhankelijk was van een parameter. Bij polynomen zullen de nulpunten ervan een speciale rol spelen.
In eenvoudige gevallen zullen we precies zo te werk kunnen gaan als in de voorbeelden van Module 14. Wanneer de matrices iets groter zijn, dan zullen we al gauw onze toevlucht moeten nemen tot numerieke benaderingen om de 'speciale gevallen' te kunnen identificeren. We geven een voorbeeld.

Voorbeeldopgave

Gegeven de matrix

$$A = \begin{bmatrix} x^3+2x^2 & 2x^3+2x^2+3x+2 & x^3+1 \\ 2x^3+x & x^3+2x+2 & 2x^3+x^2+3x+2 \\ x^3+3x^2+x & x^3+x^2+x+3 & 3x^3+3 \end{bmatrix}.$$

[39] Zie voor achtergronden van de hier behandelde onderwerpen bijvoorbeeld J.W. Polderman en J.C. Willems, *Introduction to Mathematical Systems Theory*, New York: Springer-Verlag, 1998.

Gevraagd wordt voor alle (complexe) waarden van x waarvoor A singulier is een basis voor $\ker A$ te berekenen.

Voorbeeldsessie

```
> with(LinearAlgebra):
> A := Matrix([[x^3+2*x^2, 2*x^3+2*x^2+3*x+2, x^3+1],
              [2*x^3+x, x^3+2*x+2, 2*x^3+x^2+3*x+2],
              [x^3+3*x^2+x, x^3+x^2+x+3, 3*x^3+3]]):
> p := Determinant(A);
```

$$p := -x + 26\,x^3 + 8\,x^2 + 32\,x^6 + 24\,x^4 + 37\,x^5 - 6\,x^9 + 4\,x^8 + 6\,x^7$$

```
> S := solve( p=0,{x} );
```

$$\begin{aligned}&S := \{x = 0\},\ \{x = \mathrm{RootOf}(\%1,\ \mathit{index} = 1)\},\\ &\{x = \mathrm{RootOf}(\%1,\ \mathit{index} = 2)\},\ \{x = \mathrm{RootOf}(\%1,\ \mathit{index} = 3)\},\\ &\{x = \mathrm{RootOf}(\%1,\ \mathit{index} = 4)\},\ \{x = \mathrm{RootOf}(\%1,\ \mathit{index} = 5)\},\\ &\{x = \mathrm{RootOf}(\%1,\ \mathit{index} = 6)\},\ \{x = \mathrm{RootOf}(\%1,\ \mathit{index} = 7)\},\\ &\{x = \mathrm{RootOf}(\%1,\ \mathit{index} = 8)\}\\ &\%1 := 1 - 26\,\_Z^2 - 8\,\_Z - 32\,\_Z^5 - 24\,\_Z^3 - 37\,\_Z^4 + 6\,\_Z^8 - 4\,\_Z^7 - 6\,\_Z^6\end{aligned}$$

Daar is exact niet veel mee te beginnen.
Dan maar numeriek:

```
> S := fsolve( p=0, {x} );
```

$$\begin{aligned}&S := \{x = -0.8812745978\},\ \{x = -0.5166771558\},\ \{x = 0.\},\\ &\{x = 0.09364843621\},\ \{x = 2.487144813\}\end{aligned}$$

Maple rekent alleen de reële nulpunten uit.

```
> S := fsolve( p=0, {x}, complex );
```

$$\begin{aligned}&S := \{x = -0.8812745978\},\ \{x = -0.5166771558\},\\ &\{x = -0.4157240082 - 1.355673739\,I\},\\ &\{x = -0.4157240082 + 1.355673739\,I\},\ \{x = 0.\},\\ &\{x = 0.09364843621\},\ \{x = 0.1576365935 - 0.8699034513\,I\},\\ &\{x = 0.1576365935 + 0.8699034513\,I\},\ \{x = 2.487144813\}\end{aligned}$$

Merk op dat de oplossingen geordend zijn naar het reele deel.
We vullen ze achtereenvolgens in en berekenen in elk van de gevallen de kern

```
> for i from 1 to nops([S]) do
  K[i] := evalf(S[i],3), NullSpace(subs(S[i],A))
  end do;
```

$$K_1 := \{x = -0.881\},\ \left\{\begin{bmatrix} -0.434875785872160469 \\ -0.223224038927036489 \\ 0.872384135176226039 \end{bmatrix}\right\}$$

$$K_2 := \{x = -0.517\},\ \left\{\begin{bmatrix} -0.309468435807354226 \\ -0.660199169332141622 \\ 0.684372956838666835 \end{bmatrix}\right\}$$

$$K_3 := \{x = -0.416 - 1.36\, I\},$$
$$\left\{\left[\begin{array}{c} 0.455009346363146328 + 0.\, I \\ -0.700445902170466095 - 0.476624118309172839\, I \\ 0.267479902118505641 - 0.0602161495237132360\, I \end{array}\right]\right\}$$

$$K_4 := \{x = -0.416 + 1.36\, I\},$$
$$\left\{\left[\begin{array}{c} 0.455009346363146328 + 0.\, I \\ -0.700445902170466095 + 0.476624118309172839\, I \\ 0.267479902118505641 + 0.0602161495237132360\, I \end{array}\right]\right\}$$

$$K_5 := \{x = 0.\}, \left\{\left[\begin{array}{c} 1. \\ 0. \\ 0. \end{array}\right]\right\}$$

$$K_6 := \{x = 0.0936\}, \left\{\left[\begin{array}{c} -0.998167057165902039 \\ -0.0172696451514664160 \\ 0.0580024598194362126 \end{array}\right]\right\}$$

$$K_7 := \{x = 0.158 - 0.870\, I\},$$
$$\left\{\left[\begin{array}{c} 0.733264359976164015 + 0.\, I \\ 0.0740498534451477647 + 0.434645210068529963\, I \\ 0.367106007032087401 - 0.364906451791331188\, I \end{array}\right]\right\}$$

$$K_8 := \{x = 0.158 + 0.870\, I\},$$
$$\left\{\left[\begin{array}{c} 0.733264359976164015 + 0.\, I \\ 0.0740498534451477647 - 0.434645210068529963\, I \\ 0.367106007032087401 + 0.364906451791331188\, I \end{array}\right]\right\}$$

$$K_9 := \{x = 2.49\}, \left\{\left[\begin{array}{c} 0.837947781383348911 \\ -0.300187120892980497 \\ -0.455775392188637552 \end{array}\right]\right\}$$

Toelichting

Det A is een polynoom met graad 9. De nulpunten ervan kunnen niet exact worden bepaald.[40] Het commando `fsolve` met de optie `complex` geeft numerieke benaderingen van de nulpunten.
Voor alle waarden van x waarvoor Det $A = 0$, dus rang$(A) < 3$, is een basis van de kern uitgerekend. We hebben daarbij een `for`-lus gebruikt. Zie §27.3 voor meer uitleg hierover. ◇

[40] In het resultaat van het `solve`-commando wordt het symbool `%1` gebruikt. In dit geval is het een verwijzings-symbool; in de laatste regel van de betreffende output staat wat het betekent. Op het scherm zult u dit symbool maar heel zelden tegenkomen.

15.2 De Bézout-identiteit; unimodulaire matrices

De *Bézout-identiteit* luidt:

> Als $p_1(x), \ldots, p_n(x)$ polynomen zijn die geen factor met graad > 0 gemeenschappelijk hebben, dan bestaan er polynomen $a_1(x), \ldots, a_n(x)$ zodat
>
> $$a_1(x)\, p_1(x) + \cdots + a_n(x)\, p_n(x) = 1 \tag{15.1}$$
>
> voor alle $x \in \mathbb{C}$.

Om, gegeven een vector $\mathbf{p}$ van polynomen, een vector $\mathbf{a}$ te vinden, maken we gebruik van *unimodulaire matrices*, dat zijn polynoommatrices die voor alle x inverteerbaar zijn, en waarvan de inverse ook een matrix van polynomen is. Een noodzakelijke en voldoende voorwaarde voor het unimodulair zijn van een matrix A is dat $\det A$ een (reëel of complex, al naar gelang de coëfficiënten van p_i) *getal* $\neq 0$ is. Het vinden van een polynoomvector $\mathbf{a}$ zodat aan (15.1) wordt voldaan, komt neer op het construeren van een unimodulaire matrix U waarvan $\mathbf{p}$ als kolom deel uitmaakt.
Immers, stel dat $\mathbf{p}$ de k^{de} kolom van U is, dus $U\,\mathbf{e_k} = \mathbf{p}$. Dan is, met A de inverse van U (dus óók een unimodulaire matrix): $A\,\mathbf{p} = \mathbf{e_k}$. Met andere woorden: $\mathbf{a_{k*}} \cdot \mathbf{p} = 1$, en dat is precies (15.1); de k^{de} rij van A is dus de gevraagde polynoomvector $\mathbf{a}$.

In de praktijk construeren we direct de matrix A door 'elementaire operaties' op de elementen van $\mathbf{p}$ toe te passen. Dat leggen we uit met een voorbeeld. Neem

$$\mathbf{p} = (x^3 + x^2 + 1,\ x - 1,\ 2x^2 + 3x)\,.$$

Kies nu een element van $\mathbf{p}$ van de laagste graad, dat is in dit geval $p_2(x) = x - 1$. We tellen dit zo vaak bij het eerste, respectievelijk het derde element van $\mathbf{p}$ op dat de eerste term van deze polynomen nul wordt. Dus $-x^2$ keer bij de eerste en $-2x$ keer bij de tweede. De nieuwe vector wordt dan

$$\begin{bmatrix} x^3 + x^2 + 1 & -x^2\,(x-1) \\ x - 1 & \\ 2x^2 + 3x & -2x\,(x-1) \end{bmatrix} = \begin{bmatrix} 2x^2 + 1 \\ x - 1 \\ 5x \end{bmatrix}.$$

We kunnen deze operatie gemakkelijk in de vorm van een matrixvermenigvuldiging noteren als $U_1\mathbf{p} = \mathbf{q_1}$, met

$$U_1 = \begin{bmatrix} 1 & -x^2 & 0 \\ 0 & 1 & 0 \\ 0 & -2x & 1 \end{bmatrix}.$$

De maximale graad van de elementen van $\mathbf{q_1}$ is kleiner dan de maximale graad van de elementen van $\mathbf{p}$, en de matrix U_1 is unimodulair (bereken de determinant door naar de tweede rij te ontwikkelen).
Op dezelfde manier kunnen we $\mathbf{q_1}$ bewerken:

$$U_2\mathbf{q_1} = U_2U_1\mathbf{p} = \mathbf{q_2}$$

waarbij U_2U_1 als product van unimodulaire matrices weer unimodulair is. Zo kunnen we doorgaan, totdat we op een eenheidsvector $\mathbf{e_i}$ uitkomen. De gevraagde matrix A is dan $U_m \ldots U_1$.

15.3 Berekening van Bézout-coëfficiënten

We laten aan de hand van een voorbeeld zien hoe Maple hulp kan bieden bij de berekening van Bézout-coëfficiënten.

Voorbeeldopgave

Gegeven de polynomen

$$\begin{aligned} r_1(x) &= 4x^3 + 3x^2 + 2x + 1, \\ r_2(x) &= 2x^2 + 2x + 2, \\ r_3(x) &= 3x^3 + 3x^2 + 2x + 1, \\ r_4(x) &= 4x^3 + 4x + 4\,. \end{aligned}$$

Bepaal polynomen $a_1(x), \ldots, a_4(x)$, zo dat

$$a_1(x)\,r_1(x) + \cdots + a_4(x)\,r_4(x) = 1$$

voor alle x.

Voorbeeldsessie

```
> with(LinearAlgebra):
> r := <4*x^3+3*x^2+2*x+1, 2*x^2+2*x+2,
       3*x^3+3*x^2+2*x+1, 4*x^3+4*x+4>:
```

We maken een kopie van r om mee verder te werken. Deze kunnen we steeds veranderen.

```
> q := Copy(r):
```

```
> map(degree,q,x);
```

$$\begin{bmatrix} 3 \\ 2 \\ 3 \\ 3 \end{bmatrix}$$

We zien dat q_2 de laagste graad heeft.
We maken U_1 door de tweede kolom in een eenheidsmatrix aan te passen.

```
> I4 := IdentityMatrix(4) + Matrix(4,4):
> U1 := Copy(I4):
  U1[1,2] := -lcoeff(q[1],x)/lcoeff(q[2],x)*
  x^(degree(q[1])-degree(q[2]));
```

$$U1_{1,2} := -2\,x$$

We moeten dus $-2\,x$ keer q_2 van q_1 aftrekken om de graad van q_1 omlaag te krijgen.
Kijken of het klopt:

```
> map(expand, U1.q);
```

$$\begin{bmatrix} -x^2 - 2\,x + 1 \\ 2\,x^2 + 2\,x + 2 \\ 3\,x^3 + 3\,x^2 + 2\,x + 1 \\ 4\,x^3 + 4\,x + 4 \end{bmatrix}$$

Dat doen we voortaan in een loop:

```
> for i in [1,3,4] do
    U1[i,2] := -lcoeff(q[i],x)/lcoeff(q[2])*
          x^(degree(q[i])-degree(q[2]))
  end do:
> U1;
```

$$\begin{bmatrix} 1 & -2\,x & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & -\frac{3\,x}{2} & 1 & 0 \\ 0 & -2\,x & 0 & 1 \end{bmatrix}$$

```
> q := map(expand, U1.q);
```

$$q := \begin{bmatrix} -x^2 - 2\,x + 1 \\ 2\,x^2 + 2\,x + 2 \\ 1 - x \\ 4 - 4\,x^2 \end{bmatrix}$$

```
> map(degree,q,x);
```

$$\begin{bmatrix} 2 \\ 2 \\ 1 \\ 2 \end{bmatrix}$$

De derde heeft de laagste graad

```
> U2 := Copy(I4):
  for i in [1,2,4] do
    U2[i,3] := -lcoeff(q[i],x)/lcoeff(q[3])*
        x^(degree(q[i])-degree(q[3]))
  end do:
> q := map(expand, U2.q);
```

$$q := \begin{bmatrix} -3\,x+1 \\ 2+4\,x \\ 1-x \\ 4-4\,x \end{bmatrix}$$

```
> map(degree,q,x);
```

$$\begin{bmatrix} 1 \\ 1 \\ 1 \\ 1 \end{bmatrix}$$

Laten we de derde maar nemen: minder breuken.

```
> U3 := Copy(I4):
  for i in [1,2,4] do
    U3[i,3] := -lcoeff(q[i],x)/lcoeff(q[3])*
       x^(degree(q[i])-degree(q[3]))
  end do:
> q := map(expand, U3.q);
```

$$q := \begin{bmatrix} -2 \\ 6 \\ 1-x \\ 0 \end{bmatrix}$$

Nu oppassen: er staat een nul in, en dan heb je

```
> degree(q[4]);
```

$$-\infty$$

Dat is handig, want $x^{(-\infty)} = 0$, dus we hoeven helemaal niet op te passen!

```
> U4 := Copy(I4):
  for i in [2,3,4] do
    U4[i,1] := -lcoeff(q[i],x)/lcoeff(q[1])*
        x^(degree(q[i])-degree(q[1]))
  end do:
> q := map(expand, U4.q);
```

$$q := \begin{bmatrix} -2 \\ 0 \\ 1 \\ 0 \end{bmatrix}$$

De laatste kan wel uit het hoofd:

```
> U5 := Copy(I4):  U5[1,3] := 2:
  q := map(expand, U5.q);
```

$$q := \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}$$

```
> A := map(expand, U5.U4.U3.U2.U1 );
```

$$A := \begin{bmatrix} 1-x & -\frac{1}{2}\,x - x^2 - \frac{3}{2}\,x^3 & 2\,x+x^2-1 & 0 \\ 3 & \frac{3}{2}\,x+1+\frac{3}{2}\,x^2 & -x-5 & 0 \\ -\frac{x}{2} & -\frac{5}{4}\,x^2-\frac{3}{4}\,x^3-\frac{3}{2}\,x & \frac{1}{2}\,x^2+1+\frac{3}{2}\,x & 0 \\ 0 & 4\,x+6\,x^2 & -4-4\,x & 1 \end{bmatrix}$$

Unimodulair?

```
> Determinant(A);
```

$$1$$

```
> a := A[3,1..-1];
```

$$a := \left[-\frac{x}{2},\ -\frac{5}{4}\,x^2 - \frac{3}{4}\,x^3 - \frac{3}{2}\,x,\ \frac{1}{2}\,x^2 + 1 + \frac{3}{2}\,x,\ 0\right]$$

Nu controleren we $a_1(x)\,r_1(x) + ... + a_4(x)\,r_4(x)$:

```
> expand(a.r);
```

$$1$$

```
> U := A^(-1);
```

$$U := \begin{bmatrix} -2\,x^3 - \frac{9}{2}\,x - \frac{7}{2}\,x^2 + 1 & 2\,x & 4\,x^3 + 3\,x^2 + 2\,x + 1 & 0 \\ -x^2 - 2\,x - 3 & 1 & 2\,x^2 + 2\,x + 2 & 0 \\ -\frac{(3\,x^2 + 6\,x + 8)\,x}{2} & \frac{3\,x}{2} & 3\,x^3 + 3\,x^2 + 2\,x + 1 & 0 \\ -2\,(x^2 + x + 2)\,x & 2\,x & 4\,x^3 + 4\,x + 4 & 1 \end{bmatrix}$$

De derde kolom is inderdaad precies r:

```
> Equal(U[1..-1,3],r);
```

$$\mathit{true}$$

Toelichting

De matrices U_i worden gemaakt door eerst een eenheidsmatrix te maken. We tellen er een nulmatrix bij op om waarden te kunnen veranderen, vergelijk §13.3 (voorbeeldsessie op blz. 168). Als q_k het element van $\mathbf{q}$ met de laagste graad is, worden daarna de elementen van de k^{de} kolom van U_i ingevuld.
Daarbij is de volgende formule gebruikt: Als we de eerste term van de polynoom $p(x) = ax^n + \cdots$ nul willen maken door er een aantal keren de polynoom $q(x) = bx^m + \cdots$ bij op te tellen ($m < n$), dan moeten we nemen:

$$p(x) - \frac{a}{b}x^{n-m}\,q(x)\,.$$

De coëfficiënten a en b krijgen we met `lcoeff`; de getallen n en m met `degree` (zie §11.4).
We gaan door totdat een eenheidsvector is bereikt. ◇

Deze procedure is niet eenduidig, dat wil zeggen dat er in het algemeen verschillende mogelijkheden voor $\mathbf{a}$ bestaan. Immers, als er twee of meer polynomen in $\mathbf{q}$ zijn met de laagste graad, dan is het willekeurig welke we daarvan kiezen om mee verder te werken, en het eindantwoord is daarvan afhankelijk.

15.4 Smith-normaalvormen

Zij A een $n \times n$-matrix van polynomen. Dan bestaan er unimodulaire matrices U en V, zodat de matrix

$$S = U\,A\,V$$

een diagonaalmatrix van polynomen is. Voor de diagonaalelementen van S geldt dat $s_i(x)$ een deler is van $s_{i+1}(x)$. De matrix S heet de *Smith-normaalvorm* van A.
Ook als A niet vierkant is, bestaat de Smith-normaalvorm S met dezelfde afmetingen als A. Hiervan zijn dan alle s_{ij} met $i \neq j$ gelijk aan nul; als A een $n \times m$-matrix is, dan is U een unimodulaire $n \times n$-matrix en V een unimodulaire $m \times m$-matrix.

15.5 Berekening van Smith-vormen

Het Maple-commando om Smith-vormen te berekenen bevindt zich in de `LinearAlgebra`-bibliotheek. De eenvoudigste vorm ervan is
`SmithForm` `SmithForm(A,x)`. Als A een matrix is van polynomen met x als variabele, dan levert dit de Smith-vorm van A af. Na een aanroep in de vorm

```
S,U,V := SmithForm( A, x, output=['S','U','V'] )
```

zijn dan bovendien de uitvoerparameters U en V de matrices geworden waarmee $U\,A\,V = S$.

Voorbeeldopgave

Bereken de Smith-normaalvorm van de matrix A van §15.1.

Voorbeeldsessie

```
> with(LinearAlgebra):
> A := Matrix([[x^3+2*x^2, 2*x^3+2*x^2+3*x+2, x^3+1],
               [2*x^3+x, x^3+2*x+2, 2*x^3+x^2+3*x+2],
               [x^3+3*x^2+x, x^3+x^2+x+3, 3*x^3+3]]):
> S := SmithForm(A, x);
```

$$S := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -\frac{4}{3}\,x^2 - 4\,x^4 - \frac{13}{3}\,x^3 - \frac{37}{6}\,x^5 - \frac{16}{3}\,x^6 - x^7 - \frac{2}{3}\,x^8 + x^9 + \frac{1}{6}\,x \end{bmatrix}$$

```
> U,V := SmithForm(A, x, output=['U','V']):
  Determinant(U), Determinant(V);
```

$$\frac{1}{6},\ -1$$

We berekenen een van de elementen van $U^{(-1)}$ om te zien of dat een polynoom is:

```
> U^(-1): %[2,2];
```

$$-\frac{2\,(4\,x+1)\,(51\,x^4+76\,x^3+42\,x^2+20\,x+9+24\,x^5)}{63}$$

```
> Equal( S, map( expand, U.A.V) );
```

$$true$$

```
> Equal( A, map( expand, U^(-1).S.V^(-1) ) );
```

$$true$$

Toelichting

We hebben de matrices U en V niet afgedrukt, maar alleen door de determinant te berekenen geverifieerd dat beide matrices unimodulair zijn. Dat betekent dat de inversen ervan ook 3×3-matrices van polynomen zijn. ◇

Als we voor A een $n \times 1$-matrix (een 'kolomvector') $\mathbf{r}$ nemen, dan zal de Smith-vorm ook een $n \times 1$-matrix zijn, met alleen het bovenste element $\neq 0$. De matrix V is dan een unimodulaire 1×1-matrix, dat wil zeggen dat hij uit alleen één *getal* $\neq 0$ bestaat. De unimodulaire U kan dan zó worden gemaakt dat we $V = 1$ krijgen. Als nu bovendien de elementen van A onderling copriem zijn, dan is het niet-nulelement van de Smith-vorm gelijk aan 1, en dan hebben we $U\,\mathbf{r} = \mathbf{e_1}$, dat wil zeggen dat de eerste rij van U precies de Bézout-coëfficiënten van $\mathbf{r}$ bevat.
Dat betekent dat we de voorbeeldopgave van §15.3 ook met de Mapleprocedure `SmithForm` kunnen oplossen.

Voorbeeldsessie

```
> with(LinearAlgebra):
> r := <4*x^3+3*x^2+2*x+1, 2*x^2+2*x+2,
        3*x^3+3*x^2+2*x+1, 4*x^3+4*x+4>:
> R := Matrix(r);
```

$$R := \begin{bmatrix} 4\,x^3+3\,x^2+2\,x+1 \\ 2\,x^2+2\,x+2 \\ 3\,x^3+3\,x^2+2\,x+1 \\ 4\,x^3+4\,x+4 \end{bmatrix}$$

```
> S := SmithForm( R, x );
```

$$S := \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

```
> U,V := SmithForm( R, x, output=['U','V'] ):
> U;
```

$$\begin{bmatrix} \frac{3}{7} + \frac{x}{7} & \frac{2}{7} - \frac{11}{14}\,x - \frac{2}{7}\,x^2 & 0 & 0 \\ 2\,x^2 + 2\,x + 2 & -1 - 4\,x^3 - 3\,x^2 - 2\,x & 0 & 0 \\ -\frac{(3\,x^3 + 3\,x^2 + 2\,x + 1)\,(3 + x)}{7} & \frac{(3\,x^3 + 3\,x^2 + 2\,x + 1)\,(-4 + 11\,x + 4\,x^2)}{14} & 1 & 0 \\ -\frac{4\,(x^3 + x + 1)\,(3 + x)}{7} & \frac{2\,(x^3 + x + 1)\,(-4 + 11\,x + 4\,x^2)}{7} & 0 & 1 \end{bmatrix}$$

```
> V;
```

$$\begin{bmatrix} 1 \end{bmatrix}$$

```
> U^(-1);
```

$$\begin{bmatrix} 4\,x^3 + 3\,x^2 + 2\,x + 1 & \frac{2}{7} - \frac{11}{14}\,x - \frac{2}{7}\,x^2 & 0 & 0 \\ 2\,x^2 + 2\,x + 2 & -\frac{3}{7} - \frac{x}{7} & 0 & 0 \\ 3\,x^3 + 3\,x^2 + 2\,x + 1 & 0 & 1 & 0 \\ 4\,x^3 + 4\,x + 4 & 0 & 0 & 1 \end{bmatrix}$$

```
> expand(U[1,1..-1].r);
```

$$1$$

Toelichting

Het commando `SmithForm` werkt alléén op matrices; het is daarom nodig om van de vector **r** eerst een 4×1-matrix R te maken.
Verder zien we dat we hier andere Bézout-coëfficiënten vinden dan in §15.3. ◇

Opgave 15.1

Gegeven de vector van polynomen

$$\mathbf{p}(x) = \left[x^2 - 3x + 2,\ x^2 - 5x + 6,\ x^3 - 5x^2 + 2x + 8\right] .$$

(a) Pas de methode van §15.2 toe op **p**. (Waarom) lukt het (niet) om een unimodulaire matrix A te vinden met $A\mathbf{p} = \mathbf{e_k}$?
(b) Controleer uw berekening met de procedure `SmithForm`;
(c) Factoriseer de elementen van **p** en verklaar de bij (a) en (b) gevonden antwoorden.

Opgave 15.2

randpoly

Een willekeurige polynoom kan worden gemaakt met het commando `randpoly`. Zo levert het commando

```
randpoly( x, coeffs=rand(-6..6), degree=4 )
```

een polynoom met graad ≤ 4 en als coëfficiënten gehele getallen tussen -6 en $+6$.

(a) Gebruik het bovenstaande commando in een indexfunctie om een willekeurige matrix te maken (kies zelf de afmeting);

(b) Bereken de Smith-vorm van deze matrix en de bijbehorende unimodulaire transformaties U en V;

(c) Verifieer dat $U\,A$ een bovendriehoeksmatrix is.